
The International Model of Alcohol Harms and Policies (InterMAHP): User’s
manual with worked example
To assist users with the application of InterMAHP to their region, the following manual will
describe all steps necessary to run the InterMAHP application, including preparation of input files
and interpretation of output files. A complete worked example of running InterMAHP in the
Canadian context is provided throughout the manual.

Step 1: Download and get to know InterMAHP
1.1 Download InterMAHP
Go to www.intermahp.cisur.ca and click “Download InterMAHP”. Save InterMAHP to the
location you prefer on your computer or server. Note that in order to run InterMAHP you will have
to have a version of SAS installed on your computer or server in order to run the InterMAHP
program. InterMAHP was built and tested using SAS 9.3 [1]. The SAS system must have base SAS,
as well as SAS/IML and SAS/AF. The application has been tested using both Windows 7 and
Windows 10.
There are seven items that make up the InterMAHP download, shown in the screenshot
below:
Figure 1: Screenshot of the downloaded InterMAHP package

In alphabetical order, these are:
1) .csv version of the consumption and prevalence input EXAMPLE spreadsheet. This is an
example only and will be replaced with your region-specific information regarding alcohol
consumption and prevalence. The program requires a .csv formatted file in order to run.

2) .xls version of the consumption and prevalence input EXAMPLE spreadsheet. The .xls
version is also provided as it is easier to read and work with.
3) InterMAHP Comprehensive Guide - the accompanying guide provides a detailed
methodological description of the process of estimating alcohol-attributable harms.
4) InterMAHP User’s Manual – this document.
5) intermahp SAS catalog – this file comprises both the user interface and the backend code
which is used to calculate attributable fractions.
6) .csv version of the relative risk input DEFAULT spreadsheet. This spreadsheet contains
virtually all the relative risk function and point estimate information for calculating
alcohol-attributable harm. The spreadsheet is loaded with the default relative risk values
except for ischaemic heart disease morbidity and mortality in males. Notably, because of
the contention in the literature regarding the cardioprotective effect, InterMAHP does not
have a default value for IHD morbidity and mortality in males. See Section 3.2.2 in the
Guide for a comprehensive discussion of this choice.
7) .xls version of the relative risk input DEFAULT spreadsheet. This is included as the .xls
version is easier to read and understand than the .csv. However, recall a .csv version is
necessary when you go to run the InterMAHP program
1.2 Familiarize yourself with InterMAHP by reading the Guide and User’s manual
By design, InterMAHP is easy to download, tailor to your region and run to output
estimates of alcohol-attributable mortality and morbidity. This said, it is the responsibility of the
user to understand the methods applied in order to estimate these harms and correctly explain
them when using InterMAHP to produce this output. The founding idea behind InterMAHP is that
it is completely and transparently specified; i.e. the methods presented in the Guide makes it
possible to completely recreate the results based on the information provided.
1.3 Verify that you have the necessary information for all regions
In order to undertake this exercise, you must first verify that you have access to all
necessary InterMAHP inputs (such as that shown in the consumption and prevalence input
EXAMPLE spreadsheet and information regarding the burden of mortality and morbidity) are
available at the regional level.

Step 2: Preparing InterMAHP input files
Two input files are required when running InterMAHP. The first, regarding consumption
and prevalence, contains information about alcohol consumption and exposure in your region. It
must be created by your team. A consumption and prevalence input EXAMPLE spreadsheet is
included with the InterMAHP download; however, it is only an example.
The second, regarding relative risks, is a summary of virtually all the relative risk
information needed to produce alcohol-attributable fractions using InterMAHP. Contrary to
consumption and prevalence, this spreadsheet is already loaded with (almost) the information
necessary for running InterMAHP. Your team must, however, choose a risk function for ischaemic
heart disease morbidity and mortality in males (see Section 3.2.2 in the Guide). It is easily
modifiable if your team would like to change any of the risk relationships, however the default
InterMAHP functions and values are provided (except for the case of IHD morbidity and mortality
in men).
2.1 Consumption and prevalence input spreadsheet
It is the responsibility of your team to produce all data and information needed in the
input spreadsheet consumption and prevalence.csv. Please read and understand Section 2 of the
Comprehensive Guide publication, which describes in detail the data elements needed in this input
file. Each of the 12 variables must be provided with the same column headings and in the order
shown in the example spreadsheet which was downloaded with your version of InterMAHP.
For the worked example, we have prepared an input file summarizing the prevalence and
consumption information in one Canadian province, British Columbia (BC), as well as Canada as
whole. Two regions (BC, Canada) for one year (2015) are included; but note that you can run any
number of regions and year through the InterMAHP program at one time. The following figure is a
screenshot of the included consumption and prevalence input EXAMPLE spreadsheet.

Figure 2: Screenshot of consumption and prevalence input EXAMPLE spreadsheet

The first five variables are self-explanatory and also described in Section 2.1 of the Guide.
Section 2.1 also comprehensively describes the information necessary in the other seven
variables. Note that these variables should be entered exactly as shown to avoid program
malfunction: for example, use the same order, column, headings, categories of genders and age
groups and ensure populations are whole numbers with no decimal places.
We provide a brief description of the calculation of the information in this spreadsheet.
While the user of InterMAHP is responsible for producing these estimates for their region of
interest, this is how they were conducted by researchers at the Canadian Institute for Substance
Use Research (CISUR). By column letter, these are:
(E) Population – population projections from July 1st, 2015 for BC and Canada were obtained from
Statistics Canada. These estimates are available online.
(F) PCC_litres_year – per capita sales estimates for BC and Canada were obtained from Statistics
Canada. These estimates are available online. Since Canadian alcohol sales does not include
alcohol made at home or in make-your-own stores, these estimates are adjusted upwards using a
Canadian-specific study on unrecorded alcohol [2].
(G) Correction_factor – set to 0.8 to match World Health Organization methods for the Global
Burden of Disease studies and Global Status Reports on Alcohol and Health (see Guide Section
3.3).
(H) Relative_consumption – calculated from the Canadian Tobacco, Alcohol and Drug Survey
(CTADS), conducted by Statistics Canada and provided to researchers as a Public Use Microdata
File (PUMF). This variable is the relative consumption in the six gender-age population subgroups.
Note: In the example spreadsheet, values have been standardized with men 15 to 34 set to
100.00. That is, men 35-65 drink 93.0475% as much as men 15 to 34 and women 65+ drink
5

29.6771% as much as men 15 to 34. These figures do not need to be standardized and can be in any
unit. For example, you could use your survey to calculate the average number of SD/day,
SD/week, SD/year or grams/day, grams/year or litres ethanol/year. Simply use the best survey
available in your region to produce estimate of relative consumption. The program uses this to
divide the per capita consumption in the entire population into per capita consumption in each
of the six gender-age population subgroups.
(I) P_LA – the prevalence of lifetime abstainers, i.e. those who have never consumed one or more
standard drinks.
(J) P_FD – the prevalence of former drinkers, i.e. those who have consumed one or more standard
drinks in their lifetime, but have not had one or more standard drinkers in the past year.
(K) P_CD – the prevalence of current drinkers, i.e. those who have consumed one or more
standard drinks in the past year.
(L) P_BD – the prevalence of binge drinkers among the population (not among drinkers), i.e. those
who have consumed at or above daily binge drinking level in the past month. Note the difference
here as all other prevalences are concerned with the past year. The binge drinking level is defined
differently across countries; that is why is can be defined dynamically on the InterMAHP interface.
Note on all prevalence values: these must be presented as proportions (i.e. 0.50) and not
percentages (i.e. 50% or 50.0). We also calculated the four necessary prevalence values using
CTADS, as was relative consumption. We note that the definition of “bingeing” in this survey is
5+/4+ Canadian standard drinks for men/women and so we will enter these values (in grams)
on the InterMAHP interface.
2.2 Relative risk input spreadsheet
The relative risk input spreadsheet summarizes virtually all the information needed
regarding relative risk for the InterMAHP application. A comprehensive description is provided in
Guide Section 2.2. Note the file chosen as input to the InterMAHP interface must be in .csv format.
The InterMAHP download provides both a .xls (for ease of viewing and interpretation) and .csv
(for program functionality) file. If you make changes in the .xls file and then save to .csv, we
recommend care in checking that the number of decimal places has been preserved correctly, as
.csv files can have issues with this. The .csv file provided preserves all values to 12 decimal places.

It is important to note that the relative risk spreadsheet contains the default RR choices for
the InterMAHP suite. However, the RR functions and categorical RR former values are easily
modifiable in the spreadsheet. If you change them in the spreadsheet, the change will follow
through to the submission of the InterMAHP program.

7

Step 3: Running the program
When the input files are sorted and you’re ready to run InterMAHP in order to calculate
alcohol-attributable fractions, please follow the following steps.
a) Place the intermahp SAS catalog (which was included in the download package) in the
following location on your computer: “C:\Users\USERNAME\Documents\My SAS Files\9.3”
where USERNAME is replaced with your user name. There are SAS system programs and
registries stored in this location, which should not be deleted. Once added, it will look
similar to this:
Figure 3: Screenshot of My SAS Files directory with InterMAHP catalog added

b) Open SAS. The InterMAHP interface is built in SAS Application Frame. In order to call the
interface and backend program, type the following string into the drop down box in the
top left of the SAS program window: “af c=sasuser.intermahp.intermahp.frame” (see
screenshot below). This will call the SAS catalog from the location it was placed in (a).
Figure 4: Screenshot of InterMAHP call routine

c) Hit enter and the InterMAHP interface will open automatically. It will look like this:

Figure 5: Screenshot of InterMAHP interface

d) Enter the required inputs on the InterMAHP interface. The InterMAHP interface has 11
required inputs in order to run and interact with the program. Please see Section 2.3 of the
Guide for a description of the inputs necessary to customize InterMAHP to your research
needs. It allows you to search for your input files and specify an output directory. It then
allows users to define three drinking groups (light, moderate and heavy) for detailed study
and allows a dynamic definition of binge drinking based on the customs and survey
information available in your region. It allows for a dynamic upper limit of consumption in
grams/day (see Guide Section 4.3) and a choice of dose-response extrapolation methods
for relative risks at high levels of daily consumption. See Guide Section 3.2.2 for a
comprehensive description of the extrapolation method choices.
Note: There are seven input boxes that come pre-populated with periods (“.”) as
placeholders. These must be deleted and replaced with numeric values for the program to
run.
Recall: Study the program restrictions described in the Guide Section 5.3. If these
restrictions are not followed, the results of the InterMAHP may be unreliable.
9

e) For the worked example using Canadian data, we have stored the two inputs
spreadsheets on the desktop. We enter the following inputs into the InterMAHP interface:
Figure 6: Screenshot of InterMAHP interface with worked example inputs

This will allow us to study the harm contribution in four drinking groups among
both women and men (former, light, moderate and heavy drinkers). For this study, I have
defined the current drinking groups based on the multiples of Canadian standard drinks,
which are 13.45g ethanol (or the amount of ethanol in a 341mL bottle of 5.0% ABV beer).
Light drinking is therefore defined as 0.03 to 13.5 g/day for women and 0.03 to 20.2 g/day
for men, moderate drinking as 13.5 to 27.0 g/day for women and 20.2 to 40.4 g/day for
men, and heavy drinking as 27.0 g/day to 250.0 g/day for women and 40.4 to 250.0 g/day
for men. Drinking groups may also be based, for example, on low risk drinking guidelines
in your region. See Guide Section 5.3 for input restrictions.
The binge definition in Canada, which is also commensurate with our available
survey information, is four Canadian standard drinks (53.8g) for women and five Canadian

standard drinks (67.3g) for men. Dynamically inputting these definitions on the InterMAHP
interface allows this definition to be followed through in the InterMAHP AAF calculations.
An upper limit of consumption of 250 g/day is used based on Canadian-specific
information about high volume drinkers (see also Guide Section 4.3).
Last, we have chosen the capped option for RR function extrapolation; see Guide
Section 3.2.2 for comprehensive discussion. This is the conservative choice when
estimating AAFs.
Note that it is not necessary to conduct analyses by drinking groups if you are
simply interested in the total burden of disease caused by alcohol, it is only an option to
increase analytical options. In this case, entering any reasonable values which do not go
against the program restrictions defined in the Guide will work. In the output, the AAF
components can simply be ignored and AAF_Total used.
f) Press “Submit and Run.” If you only have several Region x Year combinations, the program
typically runs in seconds; however, depending on the number of regions and year for
which AAFs are calculated, it may take longer. For example, we have just completed
calculating AAFs for a national project where we have 14 regions x 10 years; this run of the
program took about 15 minutes as each of the two AAF output files (described below) has
19,320 rows.
g) Three output files, described in the following section, are created by the InterMAHP
program and output to the location you chose on the InterMAHP interface.

Step 4: Interpreting InterMAHP output
When the InterMAHP program is successfully run, it will create three output files
containing alcohol-attributable fractions for morbidity and mortality, as well as a significant
amount of information regarding the calculation, Gamma distribution and prevalences of
drinking in your defined drinking groups.
4.1 AAF output files for morbidity and mortality AAFs
These two files are named InterMAHP_AAFs_morbidity (containing all morbidity AAFs)
and InterMAHP_AAFs_mortality (containing all mortality AAFs). These files contain all partial
alcohol-attributable fractions calculated by InterMAHP based on your region’s specified
consumption and prevalence information. The first six variables are input system variables:
Region, Year, Gender, Age_Group, IM (number) and Condition. The following four variables are
subsets of the total AAF (one each for former drinkers, light drinkers, moderate drinkers and heavy
drinkers as defined by you on the InterMAHP interface). The final column is the total AAF, which is
also the sum of the four components.
For each Region x Year, there will are 138 AAFs (23 condition/gender/outcome groups by
two genders by three age groups). To apply these to the estimated total harm in each of these
categories, it is wise to structure your healthcare morbidity data or vital statistics mortality data in
the same way (i.e. counts by Region x Year x Gender x Age_Group) for simpler
merging/combination with these AAFs files.
The screenshot below shows a two condition subset of the InterMAHP_AAFs_morbidity
file for our worked example, showing AAFs for liver cancer and pancreatic cancer.

Figure 7: Screenshot of a subset of the output file InterMAHP_AAFs_morbidity

The most often used analysis is then to use the AAF_Total variable to calculate the
alcohol-attributable harm for each condition in your region. For example, looking at the first line
(male, 15 to 34) we see that 20.50% (0.205012) of liver cancers which occurred among this
population were attributable to alcohol. This will allow you calculate you the total alcoholattributable morbidity and mortality in your region by applying the AAFs to the count in each of
these categories.
InterMAHP also allows us to ask more nuanced questions, particularly around the study of
drinking groups. In our worked example and again for the first line (liver cancer in males 15 to 34),
we can analyze what proportion of alcohol-attributable harm is experienced by each of four
drinking groups (former, light, moderate and heavy). Using our definitions inputted into the
InterMAHP interface in a previous step and analyzing the first line, we can now make calculations
which allow us to state:
a) 62.1% (0.127365/0.205012) of the alcohol-attributable (AA) liver cancer
hospitalizations in BC in men aged 15 to 34 are experienced by those who drink, on
average, three Canadian drinks or more per day (40.4g/day or more)
b) Among men aged 15 to 34, former drinkers experience a greater proportion of BC
hospitalizations due to liver cancer (13.2%) than do light drinkers (9.4%), where light
drinkers drink, on average, less than 1.5 Canadian standard drinks per day.
Of course, there are many more research questions which can be answered and often
these figures will be rolled up to totals within condition categories, or among region, or years

before drawing conclusions. But these are some examples of the questions which can now be
asked using InterMAHP.
4.2 AAF output file: prevalences, consumption and Gamma information
A third file, named InterMAHP_prev_cons_output, provides contextual information that
will be useful when writing up the results of your analysis, see screenshot below:
Figure 8: Screenshot of a subset of the output file InterMAHP_prev_cons_output

The information presented in this file refers to the six population subgroups and so there
are six row for each Region x Year (in contrast to the AAF files which are also broken out by
condition). The first five columns are information copied from your input file.
The other columns provide the following information:
(F) PCC_among_drinkers – this is the per capita consumption (in grams/day) among drinkers in
each of the six gender-age subgroups. For example, from above, male drinkers aged 15 to 34
consume an average of 35.2 g/day (or about 2.6 Canadian standard drinks).
(G) Gamma_shape and (H) Gamma_scale – these variables are provided so you can graph and
manipulate the calculated Gamma distribution used in each subgroup. This is useful for
presentations or depictions of how the Gamma distribution is composed with the various relative
risk functions.
(I) P_LA and (K) P_FD – the prevalences of lifetime abstainers and former drinkers, copied from
your input file
(K) P_LD – the calculated prevalence of light drinkers, based on the drinking group definitions
input on the InterMAHP interface. From the first line of our worked example above, we can see
that 40.7% of males aged 15 to 34 in BC are defined as light drinkers (current drinkers who
consume less than 20.2 g/day).
14

(L) P_MD – the calculated prevalence of moderate drinkers, based on the definitions provided. We
see that 17.3% of males aged 15 to 34 in BC drink, on average, between 1.5 and 3.0 Canadian
standard drinks per day (20.2 to 40.4 g/day).
(M) P_HD – the calculated prevalence of heavy drinkers, based on the definitions provided.
We see that 25.3% of males aged 15 to 34 in BC drink more than 3.0 Canadian standard drinks per
day.
Note: P_LA + P_FD + P_LD + P_MD + P_HD = 1.00 by definition. It is possible that, due to
rounding, it will be very close by not exactly one.
(N) Extrapolation_method – this is a reminder of the extrapolation method chosen for this
analysis.

4.3 Additional steps
It is important to recall these possibly additional steps once the InterMAHP output files are
created:
1) Replace InterMAHP-generated AAFs for injury categories (motor vehicle collisions,
unintentional injuries and intentional injuries) with region-specific direct AAFs wherever
possible. See Guide Section 1.6. In our worked example, we replace motor vehicle collision
AAFs with Direct AAFs calculated from Canada- and province-specific information taken
from coroner’s reports and roadside BAC testing.
2) Assign the AAFs for liver cirrhosis to the oesophageal varices category. AAFs for condition
liver cirrhosis, InterMAHP number (6).(2) are assigned to condition oesophageal varices,
InterMAHP number (5).(7), by subgroup.
3) Recall the considerations regarding oesophageal cancer, see Guide Section 1.4.1.
4) Assign AAF=1.0 to all other condition categories (as these are 100% attributable
conditions). All other conditions which do not appear in the two AAF output files are 100%
attributable by definition and therefore have their AAFs set to 1.0.
5) Calculate alcohol harms in your region by multiplying AAFs by the enumerated counts of
morbidity and mortality. See Appendix A for example SAS coding of how to assign
alcohol-related conditions to a Canadian database of hospitalizations.
6) Recall the citation and referencing guidelines for InterMAHP, described at the beginning
of the Guide and User’s manual and in the Guide Section 5.3.
15

Appendix A:
Example SAS code used to assign alcohol-related conditions to Canadian hospitalization data
/* BEGIN ALCOHOL CONDITIONS MODULE */
data Conditions_Added;
set Input_File;
/* Note: For Canadian files, DIAG_CODE_1 is the primary diagnosis and DIAG_CODE_2-25 are secondary diagnoses */
array Diag{25} DIAG_CODE_1-DIAG_CODE_25;
format Condition_Alcohol $100.;
format IM $12. ;
Condition_Alcohol="";
/* (1) Communicable diseases */
if "A15"<=substr(Diag{1},1,3)<="A19" then do;
Condition_Alcohol="Tuberculosis"; IM="(1).(1)";
end;
else if "B20"<=substr(Diag{1},1,3)<="B24" or substr(Diag{1},1,3)="Z21" then do;
Condition_Alcohol="HIV"; IM="(1).(2)";
end;
else if "J09"<=substr(Diag{1},1,3)<="J22" then do;
Condition_Alcohol="Lower respiratory tract infections"; IM="(1).(3)";
end;
/* (2) Cancer */
else if "C00"<=substr(Diag{1},1,3)<="C05" or "C08"<=substr(Diag{1},1,3)<="C10" or "C12"<=substr(Diag{1},1,3)<="C14" or substr(Diag{1},1,4)="D000" then
do;
Condition_Alcohol="Oral cavity and pharynx cancer"; IM="(2).(1)";
end;
else if substr(Diag{1},1,3)="C15" or substr(Diag{1},1,4)="D001" then do;
Condition_Alcohol="Oesaphageal cancer"; IM="(2).(2)";
end;
else if "C18"<=substr(Diag{1},1,3)<="C21" or "D010"<=substr(Diag{1},1,4)<="D014" then do;
Condition_Alcohol="Colorectal cancer"; IM="(2).(3)";
end;

else if substr(Diag{1},1,3)="C22" or substr(Diag{1},1,4)="D015" then do;
Condition_Alcohol="Liver cancer"; IM="(2).(4)";
end;
else if substr(Diag{1},1,3)="C25" or substr(Diag{1},1,4)="D017" then do;
Condition_Alcohol="Pancreatic cancer"; IM="(2).(5)";
end;
else if substr(Diag{1},1,3)="C32" or substr(Diag{1},1,4)="D020" then do;
Condition_Alcohol="Laryngeal cancer"; IM="(2).(6)";
end;
else if substr(Diag{1},1,3) in ("C50","D05") then do;
Condition_Alcohol="Breast cancer"; IM="(2).(7)";
end;
/* (3) Endocrine Condition_Alcohols */
else if substr(Diag{1},1,3) in ("E11","E13","E14") then do;
Condition_Alcohol="Diabetes"; IM="(3).(1)";
end;
else if substr(Diag{1},1,4)="E244" then do; *NOTE: This condition is very rare (at least in Canada);
Condition_Alcohol="Alcohol-induced pseudo-Cushing's syndrome"; IM="(3).(2)";
end;
/* (4) Neuropsychiatric Condition_Alcohols */
else if substr(Diag{1},1,4)="F100" or "F103"<=substr(Diag{1},1,4)<="F109" then do;
Condition_Alcohol="Alcoholic psychoses"; IM="(4).(1)";
end;
else if substr(Diag{1},1,4)="F101" then do;
Condition_Alcohol="Alcohol abuse"; IM="(4).(2)";
end;
else if substr(Diag{1},1,4)="F102" then do;
Condition_Alcohol="Alcohol dependence"; IM="(4).(3)";
end;
else if substr(Diag{1},1,4)="G312" then do;
Condition_Alcohol="Degeneration of nervous system due to alcohol"; IM="(4).(4)";
end;
else if "G40"<=substr(Diag{1},1,3)<="G41" then do;
Condition_Alcohol="Epilepsy"; IM="(4).(5)";
end;
else if substr(Diag{1},1,4)="G621" then do;
Condition_Alcohol="Alcoholic polyneuropathy"; IM="(4).(6)";
end;

else if substr(Diag{1},1,4)="G721" then do;
Condition_Alcohol="Alcoholic myopathy"; IM="(4).(7)";
end;
/* (5) Cardiovascular Condition_Alcohols */
else if "I10"<=substr(Diag{1},1,3)<="I15" then do;
Condition_Alcohol="Hypertension"; IM="(5).(1)";
end;
else if "I20"<=substr(Diag{1},1,3)<="I25" then do;
Condition_Alcohol="Ischaemic heart disease"; IM="(5).(2)";
end;
else if substr(Diag{1},1,4)="I426" then do;
Condition_Alcohol="Alcoholic cardiomypathy"; IM="(5).(3)";
end;
else if "I47"<=substr(Diag{1},1,3)<="I49" then do;
Condition_Alcohol="Atrial fibrillation and cardiac arrhythmia"; IM="(5).(4)";
end;
else if "I60"<=substr(Diag{1},1,3)<="I62" or "I690"<=substr(Diag{1},1,4)<="I692" then do;
Condition_Alcohol="Haemorrhagic stroke"; IM="(5).(5)";
end;
else if "I63"<=substr(Diag{1},1,3)<="I67" or "I693"<=substr(Diag{1},1,4)<="I694" then do;
Condition_Alcohol="Ischaemic stroke"; IM="(5).(6)";
end;
else if substr(Diag{1},1,3)="I85" then do;
Condition_Alcohol="Oesophageal varices"; IM="(5).(7)";
end;
/* (6) Digestive Condition_Alcohols */
else if substr(Diag{1},1,4)="K292" then do;
Condition_Alcohol="Alcoholic gastritis"; IM="(6).(1)";
end;
else if substr(Diag{1},1,3) in ("K70","K74") then do;
Condition_Alcohol="Liver cirrhosis"; IM="(6).(2)";
end;
else if substr(Diag{1},1,4) in ("K850","K851") or substr(Diag{1},1,4) in ("K858","K859") then do;
Condition_Alcohol="Acute pancreatitis"; IM="(6).(3)";
end;
else if "K861"<=substr(Diag{1},1,4)<="K869" then do;
Condition_Alcohol="Chronic pancreatitis"; IM="(6).(4)";
end;

else if substr(Diag{1},1,4) in ("K852","K860") then do;
Condition_Alcohol="Alcohol-induced pancreatitis"; IM="(6).(5)";
end;
/* Optional module */
/* AAFs for catgeory (A) are not calculated by InterMAHP, as maternal consumption is needed (different data sources) */
/* However, for a Canadian project, we imported direct AAFs from the US Center for Disease Control and Prevention */
/* (A) Condition_Alcohols arising during pregnancy */
else if substr(Diag{1},1,4)="Q860" then do;
Condition_Alcohol="Fetal alcohol syndrome"; IM="(A).(3)";
end;
else if substr(Diag{1},1,4) in ("P043","O354") then do;
Condition_Alcohol="Complication of pregnancy by maternal use of alcohol"; IM="(A).(1)";
end;
else if "P05"<=substr(Diag{1},1,3)<="P07" then do;
Condition_Alcohol="Low birth weight"; IM="(A).(2)";
end;
/* REVAMPED POISONINGS AND INJURIES SECTION */
/* Injuries and poisoning are different, because we are no longer only concerned with primary diagnosis, so we need nested coding */
/* A. Poisoning by alcohol */
/* Accidental and self-poisoning by alcohol are divided in alcohol and NB (non-beverage) alcohol */
else if substr(Diag{1},1,3) in ("T51") then do; /* Alcohol poisoning exists as MRDx, now check for intent - it is accidental unless specified as intentional */
IM="(8).(5)";
if substr(Diag{1},1,4) in ("T511","T512","T513","T518") then do; /* NB alcohol */
Condition_Alcohol="Accidental poisoning by NB alcohol";
do i = 1 to 25 while (Condition_Alcohol="Accidental poisoning by NB alcohol");
if substr(Diag{i},1,3)="X65" then do;
Condition_Alcohol="Intentional self-poisoning by NB alcohol"; IM="(9).(2)"; /* Intentional code X65 */
end;
end;
end;
else do;
Condition_Alcohol="Accidental poisoning by alcohol";
do i = 1 to 25 while (Condition_Alcohol="Accidental poisoning by alcohol");
if substr(Diag{i},1,3)="X65" then do;
Condition_Alcohol="Intentional self-poisoning by alcohol"; IM="(9).(2)"; /* Intentional code X65 */
end;
end;

end;
end;
/* B. Poisoning by substances other than alcohol */
else if "T36"<=substr(Diag{1},1,3)<="T50" or "T52"<=substr(Diag{1},1,3)<="T65" or substr(Diag{1},1,3) in ("T96","T97") then do; /*Poisoning there, now
intent*/
Condition_Alcohol="Accidental poisoning by substances other than alcohol"; IM="(8).(4)";
do i = 1 to 25 while (Condition_Alcohol="Accidental poisoning by substances other than alcohol");
if "X60"<=substr(Diag{i},1,3)<="X64" or "X66"<=substr(Diag{i},1,3)<="X69" then do;
Condition_Alcohol="Intentional self-poisoning by substances other than alcohol"; IM="(9).(1)";
end;
end;
end;
/* C. All other injuries can now be treated, don't need to consider intent - checked all records and (other than poisoning) intent is mutually-exclusive */
/* For injury - non-poisoning, we take the first external cause code (MRDx can be anything not listed previously, but they typically begin with "S" or "T") */
else do; /* ALL OTHER RECORDS THAT HAVE NOT ALREADY BEEN ASSIGNED A CONDTION CATEGORY (i.e. MRDx does not match any listed above) */
/* (7) Injuries - motor vehicle collisions */
do i=1 to 25 while (Condition_Alcohol="");
if substr(Diag{i},1,4) in ("V021","V029","V031", "V039", "V041", "V049", "V092","V093")
or "V123"<=substr(Diag{i},1,4)<="V129"
or "V133"<=substr(Diag{i},1,4)<="V139"
or "V143"<=substr(Diag{i},1,4)<="V149"
or substr(Diag{i},1,4) in ("V194","V195","V196","V199")
or "V203"<=substr(Diag{i},1,4)<="V209"
or "V213"<=substr(Diag{i},1,4)<="V219"
or "V223"<=substr(Diag{i},1,4)<="V229"
or "V233"<=substr(Diag{i},1,4)<="V239"
or "V243"<=substr(Diag{i},1,4)<="V249"
or "V253"<=substr(Diag{i},1,4)<="V259"
or "V263"<=substr(Diag{i},1,4)<="V269"
or "V273"<=substr(Diag{i},1,4)<="V279"
or "V283"<=substr(Diag{i},1,4)<="V289"
or substr(Diag{i},1,4) in ("V294","V295","V296","V299")
or "V304"<=substr(Diag{i},1,4)<="V309"
or "V314"<=substr(Diag{i},1,4)<="V319"
or "V324"<=substr(Diag{i},1,4)<="V329"
or "V334"<=substr(Diag{i},1,4)<="V339"
or "V344"<=substr(Diag{i},1,4)<="V349"
or "V354"<=substr(Diag{i},1,4)<="V359"

or "V364"<=substr(Diag{i},1,4)<="V369"
or "V374"<=substr(Diag{i},1,4)<="V379"
or "V384"<=substr(Diag{i},1,4)<="V389"
or substr(Diag{i},1,4) in ("V394","V395","V396","V399")
or "V404"<=substr(Diag{i},1,4)<="V409"
or "V414"<=substr(Diag{i},1,4)<="V419"
or "V424"<=substr(Diag{i},1,4)<="V429"
or "V434"<=substr(Diag{i},1,4)<="V439"
or "V444"<=substr(Diag{i},1,4)<="V449"
or "V454"<=substr(Diag{i},1,4)<="V459"
or "V464"<=substr(Diag{i},1,4)<="V469"
or "V474"<=substr(Diag{i},1,4)<="V479"
or "V484"<=substr(Diag{i},1,4)<="V489"
or substr(Diag{i},1,4) in ("V494","V495","V496","V499")
or "V504"<=substr(Diag{i},1,4)<="V509"
or "V514"<=substr(Diag{i},1,4)<="V519"
or "V524"<=substr(Diag{i},1,4)<="V529"
or "V534"<=substr(Diag{i},1,4)<="V539"
or "V544"<=substr(Diag{i},1,4)<="V549"
or "V554"<=substr(Diag{i},1,4)<="V559"
or "V564"<=substr(Diag{i},1,4)<="V569"
or "V574"<=substr(Diag{i},1,4)<="V579"
or "V584"<=substr(Diag{i},1,4)<="V589"
or substr(Diag{i},1,4) in ("V594","V595","V596","V599")
or "V604"<=substr(Diag{i},1,4)<="V609"
or "V614"<=substr(Diag{i},1,4)<="V619"
or "V624"<=substr(Diag{i},1,4)<="V629"
or "V634"<=substr(Diag{i},1,4)<="V639"
or "V644"<=substr(Diag{i},1,4)<="V649"
or "V654"<=substr(Diag{i},1,4)<="V659"
or "V664"<=substr(Diag{i},1,4)<="V669"
or "V674"<=substr(Diag{i},1,4)<="V679"
or "V684"<=substr(Diag{i},1,4)<="V689"
or substr(Diag{i},1,4) in ("V694","V695","V696","V699")
or "V704"<=substr(Diag{i},1,4)<="V709"
or "V714"<=substr(Diag{i},1,4)<="V719"
or "V724"<=substr(Diag{i},1,4)<="V729"

or "V734"<=substr(Diag{i},1,4)<="V739"
or "V744"<=substr(Diag{i},1,4)<="V749"
or "V754"<=substr(Diag{i},1,4)<="V759"
or "V764"<=substr(Diag{i},1,4)<="V769"
or "V774"<=substr(Diag{i},1,4)<="V779"
or "V784"<=substr(Diag{i},1,4)<="V789"
or substr(Diag{i},1,4) in ("V794","V795","V796","V799")
or substr(Diag{i},1,4) in ("V803", "V804", "V805")
or substr(Diag{i},1,4)="V811"
or substr(Diag{i},1,4)="V821"
or "V830"<=substr(Diag{i},1,4)<="V834"
or "V840"<=substr(Diag{i},1,4)<="V844"
or "V850"<=substr(Diag{i},1,4)<="V854"
or substr(Diag{i},1,4) in ("V860","V861","V863")
or "V870"<=substr(Diag{i},1,4)<="V879"
or substr(Diag{i},1,4) in ("V892","V893","V899","Y850") then do;
Condition_Alcohol="Motor vehicle collisions"; IM="(7).(1)";
end;
else if substr(Diag{i},1,1)="V" then do;
Condition_Alcohol="Other unintentional injuries"; IM="(8).(6)";
end;
/* (8) Unintentional injuries */
else if "W00"<=substr(Diag{i},1,3)<="W19" or substr(Diag{i},1,3)="Y30" then do;
Condition_Alcohol="Falls"; IM="(8).(1)";
end;
else if "W65"<=substr(Diag{i},1,3)<="W74" or substr(Diag{i},1,3)="Y21" then do;
Condition_Alcohol="Drowning"; IM="(8).(2)";
end;
else if "X00"<=substr(Diag{i},1,3)<="X09" or substr(Diag{i},1,3)="Y26" then do;
Condition_Alcohol="Fires"; IM="(8).(3)";
end;
else if "X40"<=substr(Diag{i},1,3)<="X44" or "X46"<=substr(Diag{i},1,3)<="X49"
or "Y10"<=substr(Diag{i},1,3)<="Y14" or "Y16"<=substr(Diag{i},1,3)<="Y19" then do;
Condition_Alcohol="Accidental poisoning by substances other than alcohol"; IM="(8).(4)";
end;
else if substr(Diag{i},1,3) in ("X45","Y15") then do;
Condition_Alcohol="Accidental poisoning by alcohol"; IM="(8).(5)";
end;

else if "W20"<=substr(Diag{i},1,3)<="W64"
or "W75"<=substr(Diag{i},1,3)<="W84"
or "X10"<=substr(Diag{i},1,3)<="X33"
or substr(Diag{i},1,3) in ("Y20","Y86","Y89")
or "Y22"<=substr(Diag{i},1,3)<="Y25"
or "Y27"<=substr(Diag{i},1,3)<="Y29"
or "Y31"<=substr(Diag{i},1,3)<="Y34"
or substr(Diag{i},1,4) in ("Y859","Y872","Y899") then do;
Condition_Alcohol="Other unintentional injuries"; IM="(8).(6)";
end;
/* (9) Intentional injuries */
else if "X60"<=substr(Diag{i},1,3)<="X64" or "X66"<=substr(Diag{i},1,3)<="X69" then do;
Condition_Alcohol="Intentional self-poisoning by substances other than alcohol"; IM="(9).(1)";
end;
else if substr(Diag{i},1,3)="X65" then do;
Condition_Alcohol="Intentional self-poisoning by alcohol"; IM="(9).(2)";
end;
else if "X70"<=substr(Diag{i},1,3)<="X84" or substr(Diag{i},1,4)="Y870" then do;
Condition_Alcohol="Other intentional self-harm"; IM="(9).(3)";
end;
else if "X85"<=substr(Diag{i},1,3)<="Y09" or substr(Diag{i},1,4)="Y871" then do;
Condition_Alcohol="Assault / homicide"; IM="(9).(4)";
end;
if substr(Diag{i},1,3)="Y35" or substr(Diag{i},1,4)="Y890" then do;
Condition_Alcohol="Other intentional injury"; IM="(9).(5)";
end;
end; drop i; /* End do loop for unintentional and intentional injuries */
end; /* End if - for all non-poisoning injuries */
/* This is a small fix if your team is looking at alcohol vs. non-beverage alcohol */
if Condition_Alcohol="Accidental poisoning by NB alcohol" then IM="(8).(5).(2)";
else if Condition_Alcohol="Accidental poisoning by alcohol" then IM="(8).(5).(1)";
else if Condition_Alcohol="Intentional self-poisoning by NB alcohol" then IM="(9).(2).(2)";
else if Condition_Alcohol="Intentional self-poisoning by alcohol" then IM="(9).(2).(1)";
run;
/* END MODULE */

